

DEX ... EFFECTS PLOT

PURPOSE

Generates a dex effects plot for a given statistic.

DESCRIPTION

A dex effects plot is a graphical method for representing a design of experiment problem. The first variable is a response variable while the remaining variables (must be at least one) represent levels of factors. The value of some user specified statistic is calculated for the response variable. The effects plot consists of:

- Vertical axis = if there are exactly 2 levels, the effect is the value of the statistic for the lower level subtracted from the value of the statistic for the higher level (this can be positive or negative). If there are more than 2 levels, the lowest value of the statistic is subtracted from the highest value of the statistic (this value is always positive);
- Horizontal axis = the factor id (i.e., 1 for factor 1, 2 for factor 2 and so on).

This plot graphically shows the following:

1. The magnitude of the maximum differences for a given statistic between levels of a factor;
2. How these maximum differences vary between factors.

SYNTAX 1

DEX <stat> EFFECTS PLOT <y> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x1> ... <xn> are a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

MEAN (or AVERAGE), MIDMEAN, MEDIAN, TRIMMED MEAN, WINDSORIZED MEAN,
 SUM, PRODUCT, SIZE (or NUMBER or COUNT), MINIMUM, MAXIMUM,
 STANDARD DEVIATION (or SD), VARIANCE,
 STANDARD DEVIATION OF MEAN (or SDM), VARIANCE OF MEAN (or VM),
 RELATIVE STANDARD DEVIATION (or RELSD),
 RELATIVE VARIANCE (or RELV or COEFFICIENT OF VARIATION),
 RANGE, MIDRANGE, LOWER HINGE, UPPER HINGE, LOWER QUARTILE, UPPER QUARTILE,
 <FIRST/SECOND/THIRD/FOURTH/FIFTH/SIXTH/SEVENTH/EIGHTH/NINTH> DECILE (or 1DEC, 2DEC,
 3DEC,4DEC,5DEC,6DEC,7DEC, 8DEC, 9DEC),
 SKEWNESS, KURTOSIS, PROPORTION,
 AUTOCORRELATION, AUTOCOVARIANCE,
 SINE FREQUENCY, SINE AMPLITUDE,
 CP, CPK, EXPECTED LOSS, PERCENT DEFECTIVE,
 SINE FREQUENCY, SINE AMPLITUDE,
 TAGUCHI SN, TAGUCHI SN+, TAGUCHI SN-, TAGUCHI SN00;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that only require a single variable to compute.

SYNTAX 2

DEX <stat> EFFECTS PLOT <y> <x> <x1> ... <xn> <SUBSET/EXCEPT/FOR qualification>

where <y> is the response variable;

<x> is a second variable used in calculating the statistic (e.g., a linear fit is computed between <y1> and <x>);

<x1> ... <xn> are a sequence of variables representing factors in a designed experiment;

<stat> is one of the following statistics:

LINEAR INTERCEPT, LINEAR SLOPE, LINEAR RESSD, LINEAR CORRELATION,

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for statistics that require two variables to compute.

EXAMPLES

DEX MEAN EFFECTS PLOT Y X1 X2

DEX SD EFFECTS PLOT Y X1 X2 X3

DEX RANGE EFFECTS PLOT Y X1 X2 X3 X4

DEX RANGE EFFECTS PLOT Y X1 TO X4

NOTE 1

This plot is normally done for a location parameter (typically the mean or median) or a spread parameter (typically the standard deviation or range). The other statistics are less often used.

NOTE 2

The TO syntax is allowed for the list of factor variables (see the EXAMPLES above).

NOTE 3

The CHARACTER, BAR, SPIKE, and LINE settings can be used to control the appearance of the plot. The trace is typically drawn with a blank line and some type of character set (the choice of character is a matter of user preference). However, you can draw the trace as a bar, a connected line, or a spike if you prefer. In any event, the user must explicitly set character and line settings (they default to all lines solid and all characters blank).

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

LINES	=	Sets the type for plot lines.
CHARACTER	=	Sets the type for plot characters
PLOT	=	Generates a data or function plot.
DEX SCATTER PLOT	=	Generates a dex scatter plot.
DEX SIGN PLOT	=	Generates a dex sign plot.
DEX ... PLOT	=	Generates a dex plot for a statistic.
DEX ... PARETO PLOT	=	Generates a Pareto dex plot for a statistic.
DEX ... YOUNDEN PLOT	=	Generates a Youden dex plot for a statistic.
DEX ... PARETO EFFECTS PLOT	=	Generates a Pareto effects dex plot for a statistic.
DEX ... ABSOLUTE EFFECTS PLOT	=	Generates an absolute effects dex plot for a statistic.
DEX ... PARE ABSO EFFECTS PLOT	=	Generates a Pareto absolute effects dex plot for a statistic.

REFERENCE

“Statistics for Experimenters,” Box, Hunter, and Hunter, Wiley and Sons, 1978.

APPLICATIONS

Design of Experiments

IMPLEMENTATION DATE

89/12

PROGRAM

```

SKIP 25
READ SHEESLE2.DAT Y PROC PLANT SPEED SHIFT PROC
.
CHARACTERS X ALL
LINES BLANK ALL
LET NFACT = 4
XLIMITS 1 NFACT
MAJOR XTIC MARK NUMBER NFACT
MINOR XTIC MARK NUMBER 0
XTIC OFFSET 1 1
XTIC LABEL FORMAT ALPHA
XTIC LABEL CONTENT PROCESS PLANT SPEED SHIFT
XILABEL FACTORS
TITLE DEX EFFECTS PLOT
.
MULTIPLY 2 2
MULTIPLY CORNER COORDINATES 0 0 100 100
YILABEL MEAN
DEX MEAN EFFECTS PLOT Y PROC PLANT SPEED SHIFT
YILABEL MEDIAN
DEX MEDIAN EFFECTS PLOT Y PROC PLANT SPEED SHIFT
YILABEL STANDARD DEVIATION
DEX SD EFFECTS PLOT Y PROC PLANT SPEED SHIFT
YILABEL RANGE
DEX RANGE EFFECTS PLOT Y PROC PLANT SPEED SHIFT
END OF MULTIPLY

```

